
=====================Spread_Spectrum_Sampling=============================== 

1) Trusting in Black_Box Spice models for transission line doesn't always work.
2) Modeling transission lines as a series of LRC networks can come close. 
3) The serial skin effect resistor RS is added to a LR network. 
4) The parallel diaelectric loss resister RP is added.
5) Mutual inductance and capacitance is modeled as the By Pass resistor RBP.  
                       _  _  _
 ___          RS      / \/ \/ \ L1   ___   1n/1cm
|IN |__/\  /\  /\_____| () () |_____|OUT| .4p/1cm
|___|    \/  \/   |           RBP|  |___|
                  |_/\  /\  /\___|___      8Ghz
                      \/  \/    |   _|_    50_Ohms
                   _/\  /\  /\__|   ___
                 _|_  \/  \/  RP     |  C1
                 ///                _|_
                                    ///

                            _-----__

                 /    /    \     /
                /    ___    \   /
               |    /   \ /  | /
               |   |   _ |__\|/
               \    \___/   //
                \     D_inch/
                 \__     __/
                    -----

A cable has dimensions where dividing it up into small LC segments does not
mean that a segment is not magnetically and electrostaticly coupled to
its adjacent segments. 

In this model, a Bypass resistor RBP seems to make things work well. Two models
are being generated to match the slow and fast cable below. Cable length 
is not specified. The models assume 10cms. 



Here is what the two models do

                                             

===================MacSpiceCode============================================
Cable_Modeling
*             RIN     V0     V1     V2     V3     V4      V5    V6     V7     V8             V20
*                        ____   ____   ____   ____   ____   ____   ____   ____          ____
*        ___/\  /\  /\__|LRCG|_|LRCG|_|LRCG|_|LRCG|_|LRCG|_|LRCG|_|LRCG|_|LRCG|...  ...|LRCG|__
*  VIN _|_    \/  \/    |____| |____| |____| |____| |____| |____| |____| |____|        |____|  |
*     /_  \               _|_    _|_    _|_    _|_    _|_    _|_    _|_    _|_           _|_   |
*    // \  \              ///    ///    ///    ///    ///    ///    ///    ///           ///   |
*    \   \//      ROUT                                                                         |
*     \___/    _/\  /\  /\_____________________________________________________________________|
*       |    _|_  \/  \/
*      _|_   ///
*      ///
*
*=========Need_A_voltage_Sources========================"
VIN       VIN    0    dc     0   AC    2
*=========MODEL_Cable1=========================
RIN       VIN    V0   50
CIN       V0     0    1p
XLRCG0    V0     V1   LRCG
XLRCG1    V1     V2   LRCG
XLRCG2    V2     V3   LRCG



XLRCG3    V3     V4   LRCG
XLRCG4    V4     V5   LRCG
XLRCG5    V5     V6   LRCG
XLRCG6    V6     V7   LRCG
XLRCG7    V7     V8   LRCG
XLRCG8    V8     V9   LRCG
XLRCG9    V9     V10  LRCG
XLRCG10   V10    V11  LRCG
XLRCG11   V11    V12  LRCG
XLRCG12   V12    V13  LRCG
XLRCG13   V13    V14  LRCG
XLRCG14   V14    V15  LRCG
XLRCG15   V15    V16  LRCG
XLRCG16   V16    V17  LRCG
XLRCG17   V17    V18  LRCG
XLRCG18   V18    V19  LRCG
XLRCG19   V19    V20  LRCG
ROUT      V20    0    50
*=========MODEL_Cable2=========================
RINB      VIN    V0B   50
CINB      V0B    0     1p
XLRCG0B   V0B    V1B   LRCGB
XLRCG1B   V1B    V2B   LRCGB
XLRCG2B   V2B    V3B   LRCGB
XLRCG3B   V3B    V4B   LRCGB
XLRCG4B   V4B    V5B   LRCGB
XLRCG5B   V5B    V6B   LRCGB
XLRCG6B   V6B    V7B   LRCGB
XLRCG7B   V7B    V8B   LRCGB
XLRCG8B   V8B    V9B   LRCGB
XLRCG9B   V9B    V10B  LRCGB
XLRCG10B  V10B   V11B  LRCGB
XLRCG11B  V11B   V12B  LRCGB
XLRCG12B  V12B   V13B  LRCGB
XLRCG13B  V13B   V14B  LRCGB
XLRCG14B  V14B   V15B  LRCGB
XLRCG15B  V15B   V16B  LRCGB
XLRCG16B  V16B   V17B  LRCGB
XLRCG17B  V17B   V18B  LRCGB
XLRCG18B  V18B   V19B  LRCGB
XLRCG19B  V19B   V20B  LRCGB
ROUTB     V20B   0    50

*=========LRCG_MODEL=========================
*                        _  _  _
*  ___          RS      / \/ \/ \ L1   ___   1n/1cm
* |IN |__/\  /\  /\_____| () () |_____|OUT| .4p/1cm
* |___|    \/  \/   |           RBP|  |___|
*                   |_/\  /\  /\___|___      8Ghz
*                       \/  \/    |   _|_    50_Ohms
*                    _/\  /\  /\__|   ___
*                  _|_  \/  \/  RP     |  C1
*                  ///                _|_
*                                     ///
.SUBCKT   LRCG  IN    OUT
 RS       IN    INS   3
 RBP      IN    OUT   200
 L1       INS   OUT   1n
 C1       OUT   0     .4p
 RP       OUT   0     8k
.ENDS     LRCG

*=========LRCG_MODELB=========================
*                        _  _  _
*  ___          RS      / \/ \/ \ L1   ___   1n/1cm
* |IN |__/\  /\  /\_____| () () |_____|OUT| .4p/1cm
* |___|    \/  \/   |           RBP|  |___|
*                   |_/\  /\  /\___|___      8Ghz
*                       \/  \/    |   _|_    50_Ohms
*                    _/\  /\  /\__|   ___
*                  _|_  \/  \/  RP     |  C1
*                  ///                _|_
*                                     ///

.SUBCKT   LRCGB IN    OUT
 RS       IN    INS   18
 RBP      IN    OUT   200
 L1       INS   OUT   1n
 C1       OUT   0     0.4p
 RP       OUT   0     3k
.ENDS     LRCGB



.control
set       pensize = 2
*AC       DECLin NUMDEC FSTART FSTOP  TRAN   TSTEP  TSTOP  TSTART TMAX   ?UIC?
ac        dec    10     .1G   5G  
plot      db(v10)       db(v10b)  ylimit -20 0

.endc

.end

